# Maintainer: Jan Alexander Steffens (heftig) <jan.steffens@gmail.com>

pkgname=tex-gyre-fonts
pkgver=2.501
pkgrel=2
pkgdesc="Substitute PostScript fonts in OpenType format"
url="http://www.gust.org.pl/projects/e-foundry/tex-gyre"
arch=(any)
license=(custom)
_zipname=tg${pkgver//./_}otf
source=($url/whole/$_zipname.zip
        $url/../licenses/GUST-FONT-LICENSE.txt)
sha256sums=('d7f8be5317bec4e644cf16c5abf876abeeb83c43dbec0ccb4eee4516b73b1bbe'
            'a746108477b2fa685845e7596b7ad8342bc358704b2b7da355f2df0a0cb8ad85')

package() {
  install -Dt "$pkgdir/usr/share/fonts/${pkgname%-fonts}" -m644 $_zipname/*.otf
  install -Dt "$pkgdir/usr/share/licenses/$pkgname" -m644 GUST-FONT-LICENSE.txt
}

# vim:set ts=2 sw=2 et:
